
 Minden adat törlése, Temp könyvtár beállítása:

-->  kill(all)$
maxima_tempdir:"c:/temp"$

 Kiindulási adatok megadása:

-->  adat:[L=3.0,F=7000.0, IE=200000.0]$

 1 Reakcióerők számítása

 A lehetséges reakcióerők a megtámasztásokból adódóan: Ax, Ay, By
A 3 statikai egyenlet ( két erő-egyensúlyi egyenlet és 1 nyomatéki
    egyenlet például az A pontra felírva):

-->  stat1:Ax=0$
stat2:Ay+By-F=0$
stat3:By*L/2-F*L=0$

 Statikai egyenletek megoldása:

-->  reak:linsolve([stat1,stat2,stat3],[Ax,Ay,By]);

 Most már felírhatóak a nyomatéki függvények mindkét szakaszon:

-->  Mh1:-Ay*x,reak;
Mh2:-Ay*x-By*(x-L/2),reak,ratsimp;

 2 Rugalmas szál differenciálegyenlete

 A rugalmas szál diffegyenlete a két szakaszon: y1''(x)=-Mh1(x)/IE és
y2''(x)=-Mh2(x)/IE. Itt most y1(x) és y2(x) függvények jelölik a
lehajlásokat.
A lehajlásfüggvények számításához kétszer kell integrálni nem
elfeledkezve az integrálási konstansokról!
Legyen fi1(x)=y1'(x) függvénnyel jelölve a szögelfordulás függvény.

-->  fi1:integrate(-Mh1/IE,x)+c1;
y1:integrate(fi1,x)+c2;

-->  fi2:integrate(-Mh2/IE,x)+c3;
y2:integrate(fi2,x)+c4;

 Az ismeretlen c1,c2,c3,c4 kiszámításához szükség van a
peremfeltételek megadására. Jelen esetben van két peremfeltételünk,
miszerint az A és B helyeken a lehajlás zérus. Valamint van két
illesztési feltételünk, hogy az y1 és y2 függvények értékei és első
deriváltjai azonosak a B helyen.

-->  pf1:ev(y1,x=0)=0$
pf2:ev(y1,x=L/2)=0$
pf3:ev(y1-y2,x=L/2)=0$
pf4:ev(fi1-fi2,x=L/2)=0$

 Megoldás számítása:

-->  cmego:linsolve([pf1,pf2,pf3,pf4],[c1,c2,c3,c4]);

 Tehát a lehajlás és szögelfordulás függvények paraméteresen és
numerikusan:

-->  y1,cmego,ratsimp;
y1num:%,adat,expand;
y2,cmego,ratsimp;
y2num:%,adat,expand;

-->  fi1,cmego,ratsimp;
fi1num:%,adat,expand;
fi2,cmego,ratsimp;
fi2num:%,adat,expand;

 3 Ábrázolás

 Ábrázoljuk mindkét lehajlásfüggvényt:

-->  wxplot2d([y1num,y2num],[x,0,L],[legend,false]),adat;

 Látható, hogy a B helyen az értékük és deriváltjaiknak értéke is
azonos. Ábrázoljuk a függvényeket csak a rájuk vonatkozó
tartományokon:

-->  ykozos: if x < L/2 then y1num else y2num, adat$

-->  wxplot2d(ykozos,[x,0,L]),adat;

 Nézzük meg hasonlóképpen a szögelfordulás függvényt:

-->  fikozos: if x < L/2 then fi1num else fi2num, adat$
wxplot2d(fikozos,[x,0,L]),adat;

 Külön ablakban megjelenítve őket:

-->  plot2d(
    ykozos,[x,0,L],
    [xlabel,"x [m]"], 
    [ylabel, "y [m]"],
    [title, "A lehajlásfüggvény"]
),adat;

-->  plot2d(
    fikozos,[x,0,L],
    [xlabel,"x [m]"], 
    [ylabel, "fi [rad]"],
    [title, "A szögelfordulás"]
),adat;

 4 Szélsőérték számítása az AB szakaszon

 Látható, hogy az AB szakaszon szélsőértéke van egy ismeretlen x0
helyen a lehajlásfüggvénynek. Ezen a helyen a szögelfordulás zérus.
Tehát x0 könnyen számítható:

-->  fi1par:fi1,cmego$
x0sol:solve(fi1par=0,x);

 A kapott megoldásokból a pozitív érték adja x0 értékét. Ezt akár
numerikusan is számíthatjuk gyökkereséssel:

-->  x0:find_root(fi1num,0,L/2),adat;

 x0 helyen a lehajlás értéke:

-->  y1,cmego,x0sol[2],fullratsimp;
%,adat,float;

 vagy:

-->  y1num,x=x0;


